Introduction to Classic Maple *by David Maslanka*

Maple is a computer algebra system designed to do mathematics. Symbolic, numerical and graphical computations can all be done with Maple. Maple's treatment of the topics of calculus is thorough. The user may compose Maple commands to carry out all of the fundamental operations that are studied in Calculus I - III, Differential Equations and Linear Algebra.

1. Starting Classic Maple

When using the Windows version of Maple, you may open the Classic Maple program by first clicking on the *Start* bar, then clicking on the *Maple* menu bar, and finally, selecting the *Classic Worksheet Maple* from the resulting Maple submenu.

CLASSIC MAPLE ICON

The Classic Maple icon shown above should appear on the Maple submenu. You may copy it and then paste it on your desktop in order to create a shortcut for opening this program.
Once the Classic Maple Worksheet window is opened, note the appearance of a **prompt sign** " >" on the first line of this worksheet and just to its right is a flashing **cursor** " | ". The prompt sign tells you that Maple is waiting for a **command**, i.e. an input.

CLASSIC MAPLE WORKSHEET WINDOW



2. Saving a Maple Worksheet

In order to name and save a Maple worksheet, begin by clicking the left hand mouse button on the word **File** which appears at the upper left corner of the *menu bar* of the Maple Worksheet Window (see figure above). From the *File menu* that results, click on the

option **Save As. . .** . At this time, a submenu will appear in the middle of the window. Type in the name that you have chosen to describe your worksheet and then click the mouse button on the rectangle **OK** . This will save your worksheet under its given name on your computer's *Drive C*.
Note: if you want to save your worksheet on a floppy disc instead, then insert your disc at this stage and follow the steps described above, but be sure to choose *Drive A* from the **Drives** Field of the **Save As. . .** submenu.

If you make some changes after naming your worksheet and decide to save the revised copy instead of the original one, then click on the floppy disc icon  located on the *tool bar* of the Maple Worksheet Window. This will immediately save the edited worksheet under the original name.

* * * Always use the left hand mouse button to click on a selection.* * *

3. Ending a Maple Session

When you want to end a Maple session, and have saved your work (if desired) then simply click on the square containing the X which is located at the extreme upper right corner of the Maple Worksheet Window.

4. Opening a Saved Maple Worksheet

In order to open a named Maple worksheet, you must first *Start Maple* as described in *Section 1*. Next, click on the word **File** located on the *menu bar*. Then click on **Open** from the resulting *File menu*. In the **Drives** Field of the *Open submenu*, select *Drive A* in case the worksheet is stored on a floppy disc, otherwise select *Drive C*. Finally, double-click on the name of the desired worksheet appearing in the large box below the field **File name** . This sequence will open the appropriate worksheet.

Note: there is a shortcut to opening a Maple worksheet. Just click on the icon displaying an open folder on the tool bar,  , then the *Open submenu* will immediately appear.

5. Defining a Maple Command

We will call any string of symbols contained between a prompt sign and a semicolon (or colon) a **command**. Commands in the Windows version of Maple always appear in red type. The semicolon " **;** " or colon " **:** " signals the end of a command.

(i) If " **;** " is used to end a command and then the **Enter** key is pressed, Maple will execute your command, if possible, and display the related outcome in blue type (for non-graphical output). If the command cannot be executed, then Maple will often respond by reprinting your command using blue characters.
(ii) If the end of your command is signaled with " **:** " and the **Enter** key is pressed, then Maple will execute the command but *will not* display the outcome.

In either case (i) or case (ii), after Maple has finished its execution, it will produce a new prompt sign and a flashing cursor on the first blank line of the worksheet.

Note:

If Maple responds to your command with an error message, then you must go back to the command line and make the necessary corrections. The cursor can be moved to the appropriate position on the command line by using the arrow keys (grouped on the right side of the keyboard), or by simply positioning the mouse arrow at the desired location on the terminal screen and clicking the left hand button one time.

The backspace key, can also be used to move the cursor. However, each time that it is pressed, the first character to the left of the cursor will be erased and all the characters on that line located to the right of the cursor will move one space leftward.

The space bar located at the bottom of the keyboard will move the cursor and all characters to its right one space rightward each time it is pressed.

---

6. Maple Symbols

Some of the elementary symbols used frequently in Maple commands are described below.

Integers: Use the keys on the top row of your computer's keyboard to enter the desired integer.

Use the hyphen character " - " also located along this row to denote a negative number.

Fractions: To enter the fraction $\frac{p}{q}$ in a command, insert the slash symbol " / " between the numbers p and q.

Decimals: Decimals are entered in the natural way.

Special Numbers:

Handwritten notation	*Maple syntax*
π	**Pi**
e	**exp(1)**
∞	**infinity**
−∞	**minus infinity**
i	**I**

Note:

Maple is case sensitive so to indicate the number π , you *must* enter capital P and small i . Similiarly, Maple *only* recognizes I as the imaginary unit. All special numbers must be input exactly as described above.

Variables: Almost any letter or string of letters and numbers can be used to name a variable. However, certain letters and strings have preassigned meanings and cannot be reassigned, e.g. **D** denotes the differential operator and cannot be used to name a variable quantity.

Numerical Operations:

Operation	*Maple syntax*
addition	+
subtraction	-
multiplication	*
division	/
exponentiation	^
factorial	**!**

Numerical Relations:

Relation	*Maple syntax*
equality	=
inequality	<>
less than	<
greater than	>
less than or equal to	<=
greater than or equal to	>=

Standard Functions:

Function	*Maple syntax*
absolute value	**abs**
square root	**sqrt**
exponential	**exp**
base e logarithm	**ln** or **log [10]**
sine	**sin**
cosine	**cos**

tangent	....	**tan**
cotangent	....	**cot**
secant	....	**sec**
cosecant	....	**csc**

Note: The trigonometric functions assume radian measure.

7. Commands for Numerical Calculations

We will write some basic commands for calculation using the symbols described above.

```
> 3+4*9;
```
$$39$$

```
> (3+4)*9;
```
$$63$$

```
> 3*exp(1)/4;
```
$$\frac{3}{4}\mathbf{e}$$

```
> (4/3)*Pi*r^3;
```
$$\frac{4}{3}\pi r^3$$

```
> 12*ln(exp(2));
```
$$24$$

```
> sin(Pi/6);
```
$$\frac{1}{2}$$

```
> tan(Pi/4);
```
$$1$$

```
> abs(-17);
```
$$17$$

```
> sqrt(25)/(2^3);
```
$$\frac{5}{8}$$

```
> 4!-2*3*4;
```
$$0$$

Note:

Every Maple worksheet is organized into **groups**, ten of which are shown in the examples above. Each **group** can be identified by a tall left bracket " [" appearing at the left edge of the window.

---

8. Commands for Decimal Representations of Real Numbers

If you would like to obtain a decimal represenation for a particular real number, then use the command name called **evalf**. For example, consider

```
> evalf(5/8);
```
$$.6250000000$$

```
> evalf(3*exp(1)/4);
```
$$2.038711371$$

Normally, Maple will calculate the decimal number to ten significant digits. However, this can be changed using the command name called **Digits**.

```
> Digits:=25;
```
$$Digits := 25$$

```
> evalf(3*exp(1)/4);
```
$$2.038711371344283926520215$$

Notes:

(i) Because Maple is case sensitive, "DIGITS": or "digits" would be viewed by this system as two different variables and *would not* effect the number of digits that Maple uses.

(ii) The symbol " **:=** " is the **assignment sign.** The value on the right is assigned to the variable on the left. If you were to leave out the " **:** ", then you would have an equation rather than an assignment.

(iii) Once "Digits" has been assigned, it affects all subsequent decimal calculations until the next command "Digits:= . . . " or the end of the sesssion. Alternatively, you may specify a number of digits within an "evalf " command affecting only that calculation, by entering:

> evalf (*number* **,** *digits* **) ;**

For example, consider the following command which calculates $\frac{3\,\mathbf{e}}{4}$ to twenty significant digits:

```
> evalf(3*exp(1)/4,20);
```

$$2.0387113713442839266$$

9. Commands involving Symbolic Calculations

We will apply some commands to polynomial and rational expressions to simplify, expand or collect like terms. First, we construct a polynomial expression:

```
> 8*x^2+16*x+6;
```

$$8\,x^2 + 16\,x + 6$$

Observe that when Maple prints the polynomial it does so in ordinary math typeset without showing the " * " between the coefficients and the variable.

```
> %/(2*x+1);
```

$$\frac{8\,x^2 + 16\,x + 6}{2\,x + 1}$$

The symbol " "% stands for " the last output ", so the result of the above command is a rational expression.

```
> simplify(%);
```

$$4\,x + 6$$

The command name **simplify** will often make a complicated expression simpler by finding cancellations. In the above command, Maple cancelled the common factor $2\,x + 1$ from numerator and denominator.

```
> expand((2*x+3)*(4*x+2));
```

$$8\,x^2 + 16\,x + 6$$

The command name **expand** is used to write an expression as a sum of simple terms. In the above command, Maple expanded the product of the two linear expressions in x.

```
> collect(2*(x^2+x)+5*x+3+3*(2*x^2+3*x+1),x);
```

$$8\,x^2 + 16\,x + 6$$

Observe that a command of the form

> collect (*expression* , *variable* **) ;**

will rewrite the given *expression* as a sum of coefficients times powers of the specified *variable*.

10. Commands to Assign Names or Values to Expressions

It is often useful to assign names to mathematical expressions. This can always be accomplished with a command of the form

> *name* **:=** *expression* **;**

which assigns to the variable called *name* the value *expression*. Here, *expression* may denote any mathematical expression.
For example, consider the following calculations involving assigned variables:

```
> y:=8*x^2+16*x+6;
```

$$y := 8\,x^2 + 16\,x + 6$$

```
> Y:=y/(2*x+1);
```

$$Y := \frac{8\,x^2 + 16\,x + 6}{2\,x + 1}$$

```
> simplify(Y);
```

$$4\,x + 6$$

```
> Z:=y*Y;
```

$$Z := \frac{(8\,x^2 + 16\,x + 6)^2}{2\,x + 1}$$

```
> Z1:=simplify(Z);
```

$$Z1 := 4\,(2\,x + 3)\,(4\,x^2 + 8\,x + 3)$$

```
> Z2:=expand(Z1);
```

$$Z2 := 32\,x^3 + 112\,x^2 + 120\,x + 36$$

A command of the form

> **subs (** $expr_{old}$ **=** $expr_{new}$ **,** *expr* **) ;**

is used to substitute $expr_{new}$ for $expr_{old}$ in the expression called *expr*. For example, consider the following commands:

```
> subs(x=2, Z1);
```

$$980$$

```
> subs(x=-3*v,Y);
```

$$\frac{72\,v^2 - 48\,v + 6}{-6\,v + 1}$$

11. Commands for Defining Functions

The simplest way to define a function in Maple is with a mapping command of the form:

> *fcn* **:=** *var* **->** *expr* **;**

which assigns the name *fcn* to the mapping of the variable *var* into the expression *expr* (where, of course, *expr* involves *var* in the case that *fcn* is to be nonconstant).
Consider the following examples:

```
> f:=x->x+abs(x);
```

$$f := x \to x + |x|$$

The function f can now be evaluated at any value of x by writing this value in parentheses.
For instance,

```
> f(5);
```

$$10$$

```
> f(-4);
```

$$0$$

```
> f(a+b);
```

$$a + b + |a + b|$$

Note the difference between the function f defined above and the expression y defined in Maple by:

```
> y:=x+abs(x);
```

$$y := x + |x|$$

To evaluate this expression y when $x = 5$, we must use the more complicated command :

```
> subs(x=5,y);
```

$$5+|5|$$

```
> simplify(%);
```

$$10$$

However, we can convert an expression into a function with a command of the form:

> *fcn* **:=** **unapply (** *expr* **,** *var* **) ;**

which assigns the name *fcn* to the mapping of the variable *var* into the expression *expr*.
For example, consider

```
> r:=unapply(y,x);
```

$$r := x \to x+|x|$$

```
> r(5);
```

$$10$$

Sometimes, we may need to represent a piecewise defined function in Maple. To illustrate one way of doing this, consider the following example. Suppose that

$$\mathrm{f}(x) = \begin{cases} x^2 & x < -1 \\ x-2 & -1-x \le 0 \text{ and } x-1 < 0 \\ x^3 & \textit{otherwise} \end{cases}$$

In order to define this function f, we enter the following:

```
> f:=x->piecewise(-1>x, x^2,-1<=x and 1>x, x-2 , x^3);
```

$$f := x \to \mathrm{piecewise}(x < -1, x^2, -1 \le x \text{ and } x < 1, x-2, x^3)$$

Observe that the first input within the **piecewise** command is the condition on the first *piece* of of the function, and the second input is the formula for that piece. You may input as many *pieces* as required. Note that the last input is the formula to be used when none of the specified conditions hold. Therefore, it is not necessary to specify conditions on the last part of the formula for f.
To check our definition, we evaluate f at x .

```
> 'f(x)'=f(x);
```

$$\mathrm{f}(x) = \begin{cases} x^2 & x < -1 \\ x-2 & -1-x \le 0 \text{ and } x < 1 \\ x^3 & \textit{otherwise} \end{cases}$$

12. Commands for Solving Equations

A command of the form

> **solve (** *equa* **,** *var* **) ;**

asks Maple to solve the equation *equa* in terms of the variable *var*. For example, consider

```
> s:=solve(x^5+12*x^3-4*x=0,x);
```

$$s := 0, -\sqrt{-6+2\sqrt{10}}, \sqrt{-6+2\sqrt{10}}, -\sqrt{-6-2\sqrt{10}}, \sqrt{-6-2\sqrt{10}}$$

Each of the four solutions listed above has a unique name. The first is denoted by s [1], the second by s [2] , . . . , and the fourth by s [4]. To verify this, enter one such name at a prompt sign.

```
> s[3];
```

$$\sqrt{-6+2\sqrt{10}}$$

Observe that we get the eappropriate value : $-\sqrt{-6+2\sqrt{10}}$ for s [3] .
Maple can often solve equations which have an infinite number of solutions. Recall how the function r was defined above and consider the command:

```
> solve(r(x)=0,x);
```

$$\text{RealRange}(-\infty, 0)$$

However, sometimes it is too complicated for Maple to express all of the solution to an equation in terms of the usual operations of algebra.

```
> Solutions:=solve(x^6+12*x^3-4*x=0,x);
```

$Solutions := 0, \text{RootOf}(\_Z^5+12\_Z^2-4, index=1), \text{RootOf}(\_Z^5+12\_Z^2-4, index=2), \text{RootOf}(\_Z^5+12\_Z^2-4, index=3), \text{RootOf}(\_Z^5+12\_Z^2-4, index=4), \text{RootOf}(\_Z^5+12\_Z^2-4, index=5)$

To find all possible values of an expression involving RootOf , we may use the command
> **allvalues (** *expr* **) ;**

```
> allvalues(Solutions[2]);
```

$$\text{RootOf}(\_Z^5+12\_Z^2-4, index=1)$$

Observe that one pair of solutions to the fifth degree polynomial equation are complex conjugates of one another. Note also that Maple has given decimal approximation to the real and imaginary parts of all solutions to ten significant digits.

Alternatively, we could have used the command
> **fsolve (** *equ* **,** *var* **) ;**
to find decimal representations for the real solutions.

```
> fsolve(x^6+12*x^3-4*x=0,x);
```

-2.2374433543362915255452 78, -.58215580580450599487520 94, 0., .57287981514823976002905 26

For a nonpolynomial equation the **solve** and **fsolve** commands will not always yield all real solutions.
In case we are interested in the particular solution to an equation which lies in the interval [a , b] , then we may try the following command :
> **fsolve (** *equ* **,** *var* **,** *a* **. .** *b* **) ;** .

```
> solve(cot(Pi*x)=0,x);
```

$$\frac{1}{2}$$

```
> fsolve(cot(Pi*x)=0,x,1..2);
```

1.500000000000000000000000

```
> fsolve(cot(Pi*x)=0,x,10..11);
```

10.50000000000000000000000

13. Commands for Plotting

I . Plotting Expressions
In order to plot an expression involving a single variable over the interval [a , b] , we use a command of the form:
> **plot (** *exp* **,** *var* **=** *a* **. .** *b* **) ;**
For example, suppose

```
> y:=3*x^4-4*x^3;
```

$$y := 3x^4-4x^3$$

```
> plot(y,x=-2..2);
```



Note that the general shape of this graph is not easily recognized due to the large values that y attains when x lies in the interval [-2 , -1]. We can restrict the range of the plotted curve to the interval [c , d] by entering :

> **plot (** *exp* **,** *var* = *a* **..** *b* , *c* . . *d* **) ;** .

```
> plot(y,x=-2..2, -1..4);
```



<u>II . Plotting Continuous Functions of One Variable</u>

We use a command of the form

> **plot (** *fcn* **,** *a* . . *b* **) ;**

to plot the function called *fcn* over the interval [a , b] .

Consider the following example.

```
> h:=t->abs(t)^(1/3)*signum(t);
```

$$h := t \rightarrow |t|^{(1/3)}\,\mathrm{signum}(t)$$

Recall that the **signum** function , denoted by *sgn* , is defined by

$$\mathrm{sgn}(t) = \begin{cases} 1 & 0 < t \\ 0 & t = 0 \\ -1 & t < 0 \end{cases}$$

Therefore, we can express the function h more simply as:

$$h(t) = t^{\left(\frac{1}{3}\right)} .$$

```
> plot(h,-10..10);
```



Analogous to the case of expressions, we may restrict the plot of the range of a function to the interval [c , d], by entering

> **plot (** *fcn* **,** $a\,..\,b$, $c\,..\,d$ **) ;** .

```
> plot(h,-10..10,-1.5 .. 1.5);
```



The option **thickness** = ***n*** (where n = 0, 1, 2, or 3) may be used in a plot command to control the width of the plot of a curve. The default thickness is 0.
For instance, the previous plot can be made to appear bolder by entering

> **plot (** *fcn* **,** $a\,..\,b$, $c\,..\,d$, *thickness =2* **) ;** .

```
> plot(h,-10..10,-1.5 .. 1.5,thickness=2);
```



III . Plotting Functions of One Variable Possessing Jump Discontinuities

Recall the piecewise defined function f from the example in *Section 11* :

```
> f:=x->piecewise(-1>x, x^2,-1<=x and 1>x, x-2 , x^3);
```

$$f := x \to \text{piecewise}(x < -1, x^2, -1 \le x \textbf{ and } x < 1, x - 2, x^3)$$

```
> plot(f,-3..2,thickness=2);
```



Because the **plot** command joins each ordered pair (x , y) that it computes to the next one, the graph appears to have vertical segments at the points of discontinuity $x = -1$ and $x = +1$.
To correct this problem, we may use the option " discont = true " in the **plot** command for an expression or a **function**.

```
> plot(f,-3..2,discont=true,thickness=2);
```



IV . Plotting Several Functions or Expressions Together

The forms of the commands for plotting several functions or expressions together are analogous to those described above in III. However, the set of functions/expressions to be plotted must be enclosed by curly braces " { } " within the **plot** command. Consider the following examples:

(*i*) *Plotting two or more expressions together*

```
> w:=3*x^4-4*x^3-12*x^2+5;
```

$$w := 3\,x^4 - 4\,x^3 - 12\,x^2 + 5$$

```
> v:=10*x^3-43*x;
```

$$v := 10\,x^3 - 43\,x$$

```
> plot({w,v},x=-2..2,thickness=2);
```



(*ii*) *Plotting two or more functions together*

```
> g:=x->x^2;
```

$$g := x \to x^2$$

```
> h:=x->x^3;
```

$$h := x \to x^3$$

```
> plot({g,h},-2..2,thickness=2);
```



(*iii*) *Plotting functions and expressions together*

```
> plot({g(x),h(x),w,v},x=-2..2,thickness=2);
```



(*iv*) *Plotting curves together with different linestyles for distinction*

```
> plot([g(x),h(x),w,v],x=-2..2,linestyle=[1,2,3,4],thickness=2);
```



<u>V. Plotting Equations in Two Variables</u>

The command name **implicitplot** is located in Maple's "plots" package and is used to graph equations for which one variable is not readily expressible as a function of the other. To read in the "plots" package, we enter the command

> **with(plots) :** .

This package only needs to be read in once during a Maple session, after which, you can use all the functions from this package.

Now the general form of the equation plotting command is given by:

> **implicitplot(** *equ* , var_1 = *a* . . *b* , = *c* . .var[1] *d* **) ;**

where *equ* denotes the equation to be plotted over the ranges [*a* , *b*] and [*c* , *d*] for the variables var_1 and var_2 respectively. Consider the following example.

```
> restart:
> equ:=x^4+y^4-4*(y^3+x^2+x^2*y)+2*x^2*y^2=0;
```

$$equ := x^4 + y^4 - 4\,y^3 - 4\,x^2 - 4\,x^2\,y + 2\,x^2\,y^2 = 0$$

```
> with(plots):
Warning, the name changecoords has been redefined
```

```
> implicitplot(equ,x=-3..3,y=-1..4,scaling=constrained,thickness=2);
```



Note that this graph is not very accurate. Clearly, (0 , 0) is a point which satisfies the equation and yet the graph does not pass through the origin. We can use the option $grid = [\ int_1\ , int_2\ ]$ to increase the grid size of the plot. The default size is [25, 25] .

```
> implicitplot(equ,x=-3..3,y=-1..4,grid=[85,85],scaling=constrained,thickness=2);
```



14. Instructions for Labeling Graphs

Consider the function f defined below:

```
> f:=x->piecewise(x<=0,x^2,x*sin(1/x));
```

$$f := x \to \text{piecewise}\left(x \le 0, x^2, x \sin\left(\frac{1}{x}\right)\right)$$

To check our definition of f, we examine the value of $f(x)$.

```
> `f(x)`=f(x);
```

$$f(x) = \begin{cases} x^2 & x \le 0 \\ x \sin\left(\frac{1}{x}\right) & otherwise \end{cases}$$

In order to label the graph of f, we use the following procedure:

```
> with(plots):
> P:=plot(f,-0.5..0.5,color=blue,thickness=2):
> Q:=textplot([0.35,0.26,`y = f(x)`]);
```

$$Q := \text{PLOT}(\text{TEXT}([.35, .26], "y = f(x)"))$$

```
> display(P,Q);
```



The command **P := . . .** assigns the name, **P**, to the graph of f. The option **color =** was used within this plotting command to specify a color for the curve $y = f(x)$ .The colon, **:**, at the end of this line tells *Maple* not to display any output at this time. The command **Q := . . .** centers the string of characters " **y = f (x)** " at the point (0.35 , 0.26) in the coordinate plane. Finally, the "plots" package was read with in:

> with (plots) :

because this package contains the **display** command. The statement [> **display (P,Q) ;** then merges the graph of f with the string " **y = f(x)** " on a single coordinate plane.

It is not hard to show that when the value of the function f is minimized, then $x = \cot\left(\frac{1}{x}\right)$.

Specifically, this occurs when $x = m$, where to 10 significant digits:

```
> m:=fsolve(x=cot(1/x),x,0.2..0.3);
```

$m := .2225481584$

Thus, the minimum function value n is equal to $f(m)$.

```
> n:=f(m);
```

$n := -.2172336282$

Suppose we wish to label the minimum point on the graph by the letter **M** . We may do this with the following command:

```
> R:=textplot([m,n-.005,`M`],align = BELOW);
```

R := PLOT(TEXT([.2225481584, -.2222336282], "M", *ALIGNBELOW*))

Note that the **align** assignment tells *Maple* to write the letter **M** directly below the point with coordinates (m , $n - .5\text{e-}2$) in the plane. We may type **LEFT**, **RIGHT**, **ABOVE**, or **BELOW** after **align =** .

Now we redraw the graph of f indicating the point **M**.

```
> display(P,Q,R);
```



15. Commands for Differentiation and Integration

Enter the command

> **D** (*fcn*) **;**

where *fcn* is the name of a function. Maple responds by calculating the derivative *function*.

```
> S:=t->t^4-t^3+t^2;
```

$$S := t \to t^4 - t^3 + t^2$$

```
> s:=D(S);
```

$$s := t \to 4\,t^3 - 3\,t^2 + 2\,t$$

Enter the command

> **diff** (*expr* , *var*) **;**

where *expr* is an expression to be differentiated with respect to the variable called *var* , and Maple returns the derivative *as an expression*.

```
> Y:=x^4-x^3+x^2;
```

$$Y := x^4 - x^3 + x^2$$

```
> y:=diff(Y,x);
```

$$y := 4\,x^3 - 3\,x^2 + 2\,x$$

Enter the command

> **int** (*expr* , *var*) **;**

to obtain an antiderivative, (no constant of integration appears), of the given expression *expr* with respect to the variable *var* .

Note that Maple returns an expression, not a function.

```
> int(Y,x);
```

$$\frac{1}{5}x^5 - \frac{1}{4}x^4 + \frac{1}{3}x^3$$

The use of the command name **Int** will write the integral in text form.

```
> Int(Y,x);
```

$$\int x^4 - x^3 + x^2\, dx$$

Thus, we may enter :

```
> Int(Y,x)=int(Y,x);
```

$$\int x^4 - x^3 + x^2\, dx = \frac{1}{5}x^5 - \frac{1}{4}x^4 + \frac{1}{3}x^3$$

To obtain the definite integral of an expression, we use a command of the form

> **int** (*expr* , *var* = *a* . . *b*) **;** .

```
> int(Y,x=0..2);
```

$$\frac{76}{15}$$

We may again use the **Int** command name to write the definite integral in text form.

```
> Int(Y,x=0..2)=int(Y,x=0..2);
```

$$\int_0^2 x^4 - x^3 + x^2\, dx = \frac{76}{15}$$

```
>
>
>
```

[>

16. Inserting a Prompt Sign Between Two Maple Commands

In order to insert a new prompt sign between two existing Maple commands, just move the cursor to the end of the first command line (i.e., just to the right of the terminal **:** or **;**) and click on the " **[>** " icon located on the top *tool bar* . This will have the desired effect of causing a prompt sign to appear between the two existing command lines.

17. Commands for Adding Text to Maple Worksheets

Maple may be used as a word processor to add text to a Maple worksheet. When the cursor is flashing just to the right of a prompt sign you may add comments in text form by clicking on the **text** icon which appears as a bold capital T in the middle of the *tool bar* . This will cause the prompt sign to disappear and all characters entered now will appear in black type. The black type indicates that these symbols are in text format. *Maple commands cannot be entered while in text mode.* After you have completed typing all your comments in the section you may return to Maple command mode by clicking on the capital sigma icon " Σ " which appears on the *tool bar* . This will cause a prompt sign to appear on the next line of the worksheet with a flashing cursor positioned just to its right. At this point, commands may be entered in the usual way and all characters will appear in red once again.

To insert mathematical symbols or expressions while in text mode, first press the space bar on the keyboard once, then click on the sigma icon " Σ " on the *tool bar* . This will cause a bold black question mark to appear inscribed in a box at the position where the cursor had been flashing. Now enter the appropriate Maple string for the desired mathematical symbol, just as you would as if Maple were in its command mode. The desired mathematical symbol will now appear inside the box that had previously contained the question mark. Finally, click on the T icon on the *tool bar* to delete the circumscribed box and to continue typing comments in text mode.

For instance, to enter the expression $\frac{x^3}{x^4 + \pi x}$ in a comment while in text mode :

1. Press the space bar once .
2. Click on the " Σ " icon .
3. Enter : x ^ 3 / (x ^ 4 + Pi * x)
4. Click on the T icon .